/***************************************************************************
 *   oboe.controllers.jml.helpers.JMLPopUp                                 *
 *   Copyright (C) 2011 by Felipe Manga                                    *
 *   lufeboma@gmail.com                                                    *
 *                                                                         *
 *   Permission is hereby granted, free of charge, to any person obtaining *
 *   a copy of this software and associated documentation files (the       *
 *   "Software"), to deal in the Software without restriction, including   *
 *   without limitation the rights to use, copy, modify, merge, publish,   *
 *   distribute, sublicense, and/or sell copies of the Software, and to    *
 *   permit persons to whom the Software is furnished to do so, subject to *
 *   the following conditions:                                             *
 *                                                                         *
 *   The above copyright notice and this permission notice shall be        *
 *   included in all copies or substantial portions of the Software.       *
 *                                                                         *
 *   THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,       *
 *   EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF    *
 *   MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. *
 *   IN NO EVENT SHALL THE AUTHORS BE LIABLE FOR ANY CLAIM, DAMAGES OR     *
 *   OTHER LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, *
 *   ARISING FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR *
 *   OTHER DEALINGS IN THE SOFTWARE.                                       *
 ***************************************************************************/

package oboe.controllers.jml.helpers
{
	import flash.display.DisplayObject;
	import flash.display.Sprite;
	import flash.events.Event;
	import flash.events.MouseEvent;
	
	import oboe.controllers.jml.JMLController;
	import oboe.core.Oboe;
	import oboe.utils.CallbackHandle;
	
	public class JMLPopUp extends JMLNode
	{
		private static var activePopUp:JMLPopUp;
		
		private var asset:Sprite;
		private var cbh:CallbackHandle;
		
		public function JMLPopUp(head:String, jml:Object, parent:JMLController)
		{
			super(head, jml, parent);
			this.asset = this.getDisplayObject() as Sprite;
		}
		
		public function slot_showPopUp( id:String ):void
		{
			if( id != this.asset.name || activePopUp == this ) return;
			if( activePopUp && activePopUp != this )
			{
				activePopUp.message("hidePopUp");
			}
			activePopUp = this;
			this.asset.visible = true;
			this.cbh = Object(Oboe.stage).addCallbackEx( MouseEvent.CLICK, onClick );
			this.cbh.ignoreCount = 1;
		}
		
		private function onClick( e:Event ):void
		{
			var p:DisplayObject = e.target as DisplayObject;
			while( p && p != this.asset )
			{
				p = p.parent;
			}
			
			if( !p ) this.slot_hidePopUp();
		}
		
		protected override function createDisplayObject(type:String):DisplayObject
		{
			var s:Sprite = new Sprite();
			s.visible = false;
			return s;
		}
		
		public function slot_hidePopUp():void
		{
			this.getDisplayObject().visible = false;
			if( activePopUp == this )
			{
				activePopUp = null;
				Object(Oboe.stage).removeCallback( this.cbh );
				this.cbh = null;
			}
		}
	}
}